CHAPTER 14 - APPLICATIONS : QED

OUTLINE of the chapter

Section 14.4 ▸

Compton scattering

$$\gamma + e \rightarrow \gamma + e$$

Recall the history;

- ❑ Arthur H. Compton won the Nobel Prize in Physics in 1927 for his 1923 discovery of the Compton effect, which demonstrated the particle nature of electromagnetic radiation.(*Wiki*)
- ❑ Now, calculate the cross section for arbitrary energies.
- ❑ Klein, O; Nishina, Y (1929). "Über die Streuung von Strahlung durch freie Elektronen nach der neuen relativistischen Quantendynamik von Dirac". *Z. Phys*. **52** (11-12): 853 and 869.

Kinematics $\gamma(k_1) + e(p_1) \rightarrow \gamma(k_2) + e(p_2)$



$s = (k_1 + p_1)^2$;
$t = (k_1 - k_2)^2$; $u = (k_1 - p_2)^2$

Lab frame =
rest frame of e_1



$$k_1^{\mu} = (k, 0, 0, k) \quad ; \quad p_1^{\mu} = (m, 0, 0, 0)$$

$$k_2^{\mu} = (\omega_2 , \mathbf{k_2}) \quad ; \quad p_2^{\mu} = (E_2 , \mathbf{p_2})$$

❖ **There are 2 Feynman diagrams.**



$$M_1 = \bar{u}_2 \, (ie\gamma^{\mu}) \, iS_F(p_1+k_1) \, (ie\gamma^{\nu}) \, u_1 \, \varepsilon_{2\mu} \, \varepsilon_{1\nu}$$

$$M_2 = \bar{u}_2 \, (ie\gamma^{\mu}) \, iS_F(p_1-k_2) \, (ie\gamma^{\nu}) \, u_1 \, \varepsilon_{2\nu} \, \varepsilon_{1\mu}$$

$$M_1 = - ie^2 \; / \; (s - m^2)$$
$$\bar{u}_2 \, \gamma^{\mu} \, (\not{p}_1 + \not{k}_1 + m) \, \gamma^{\nu} \, u_1 \;\; \varepsilon_{2\mu} \, \varepsilon_{1\nu}$$

$$M_2 = - ie^2 \; / \; (u - m^2)$$
$$\bar{u}_2 \, \gamma^{\mu} \, (\not{p}_1 - \not{k}_2 + m) \, \gamma^{\nu} \, u_1 \;\; \varepsilon_{2\nu} \, \varepsilon_{1\mu}$$

❖ Implications of gauge invariance

If there is an external photon (initial or final) then the matrix element looks like this

$$\mathcal{M} = \varepsilon_\mu M^\mu \; .$$

We can prove from gauge invariance that

$$k_\mu M^\mu = 0 \; .$$

Proof *(Maiani)*

The polarization vector comes from

$$\int \pounds_{int}\, d^4x = \int e\, j^\mu(x)\, A_\mu(x)\, d^4x \; ;$$

from the expansion of $A_\mu(x)$ in plane waves, there is a term (a or a+) $e^{ik.x}\, \varepsilon_\mu$.

Now consider a gauge transformation ,

$$A_\mu(x) \rightarrow A'_\mu(x) = A_\mu(x) + \partial_\mu \Lambda(x),$$

with $\Lambda(x) = e^{ik.x}\, C$;

$$A'_\mu(x) = (\text{ a or a+ })\, e^{ik.x}\, [\, \varepsilon_\mu + ik_\mu C \,].$$

The matrix element must be gauge invariant, which requires $k_\mu M^\mu = 0$. *(qed)*

Or, another proof

Local charge conservation implies the continuity equation, $\partial_\mu j^\mu(x) = 0$.

Or, in momentum space, $k_\mu \hat{j}^\mu(k) = 0$.

For example, $k_\mu \bar{u}(p+k)\, \gamma^\mu\, u(p) = 0$.

So $k_\mu M^\mu = 0$. *(qed)*

Now consider $|\mathcal{M}|^2$ where $\mathcal{M} = \varepsilon_\mu M^\mu$.

$$|\mathcal{M}|^2 = \varepsilon_\mu M^\mu \varepsilon_\alpha (M^\alpha)^* . \qquad \bigstar$$

For unpolarized scattering we average over initial polarizations and sum over final polarizations. In either case we have

$$\sum_{r=1,2} \varepsilon_{r,\mu} \varepsilon_{r,\alpha} = - g_{\mu\alpha}$$
$$- [k_\mu k_\alpha - (k.n)(k_\mu n_\alpha + n_\alpha k_\mu)] /(k.n)^2 ;$$

$n^\mu = (1,0,0,0)$.

The second line gives 0 in $\bigstar$.

Maiani: "when multiplying by gauge invariant amplitudes, the sum over photon polarizations may be replaced by $- g_{\mu\alpha}$."

❖ Calculate $|M_s|^2$ and $|M_u|^2$

$$M_s = - ie^2 \;/\; (s - m^2)$$
$$\bar{u}_2 \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu u_1 \;\; \varepsilon_{2\mu} \varepsilon_{1\nu}$$

$$|M_s|^2 = \frac{e^4}{(s-m^2)^2} \bar{u}_2 \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu u_1 \;\; \epsilon_{2\mu} \epsilon_{1\nu}$$
$$\bar{u}_1 \gamma^\alpha (\not{p}_1 + \not{k}_1 + m) \gamma^\beta u_2 \;\; \epsilon_{2\beta} \epsilon_{1\alpha}$$

For unpolarized scattering,

$$\overline{|M_s|^2} = \frac{e^4}{(s-m^2)^2} \frac{1}{4} \mathrm{Tr}\, \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu u_1 \bar{u}_1 \to \frac{\not{p}_1 + m}{2m}$$
$$\gamma^\alpha (\not{p}_1 + \not{k}_1 + m) \gamma^\beta u_2 \bar{u}_2 \to \frac{\not{p}_2 + m}{2m}$$

$$\times g_{\nu\alpha} g_{\mu\beta}$$

$$\overline{|M_s|^2} = \frac{e^4}{(s-m^2)^2} \frac{1}{16 m^2}$$
$$\mathrm{Tr}\{ \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu (\not{p}_1 + m) \gamma_\nu (\not{p}_1 + \not{k}_1 + m) \gamma_\mu (\not{p}_2 + m) \}$$

$\overline{|M_u|^2}$ is the same with $s \rightleftarrows u$.

$$|\mathbf{M_s}|^2 = \text{coefficient} * \text{traces} = \lambda_{ss}\,\xi_{ss}$$

$$\lambda_{ss} = e^4 / (s - m^2)^2 / (16m^2)$$

$$\xi_{ss} = \mathrm{Tr}\,\{\, \gamma^\mu (p_1+k_1+m)\, \gamma^\nu (p_1+m)\, \gamma_\nu (p_1+k_1+m)\, \gamma_\mu (p_2+m) \,\}$$

$$|\mathbf{M_u}|^2 = \text{coefficient} * \text{traces} = \lambda_{uu}\,\xi_{uu}$$

$$\lambda_{uu} = e^4 / (u - m^2)^2 / (16m^2)$$

$$\xi_{uu} = \mathrm{Tr}\,\{\, \gamma^\mu (p_1 - k_2+m)\, \gamma^\nu (p_1+m)\, \gamma_\nu (p_1-k_2+m)\, \gamma_\mu (p_2+m) \,\}$$

USE FEYNCALC to evaluate

$$|M_s|^2 = \lambda_{ss}\,\xi_{ss} \quad \text{and} \quad |M_u|^2 = \lambda_{uu}\xi_{uu}$$

Compton e(p1) + γ(k1) → e(p2) + γ(k2)

```
In[1]:= $LoadFeynArts = False;
        << HighEnergyPhysics`FeynCalc`
        Loading FeynCalc from /home/stump/.Mathematica/Appl
        FeynCalc 8.2.0 For help, type ?FeynCalc, open FeynCa

In[3]:= (* scalar products *)
        ScalarProduct[p1, p1] = m^2;
        ScalarProduct[p2, p2] = m^2;
        ScalarProduct[k1, k1] = 0;
        ScalarProduct[k2, k2] = 0;

In[7]:= (* Mandelstam variables *)
        ScalarProduct[p1, k1] = (s - m^2) / 2;
        ScalarProduct[p2, k2] = (s - m^2) / 2;
        ScalarProduct[p1, p2] = (2 m^2 - t) / 2;
        ScalarProduct[k1, k2] = (-t) / 2;
        ScalarProduct[p1, k2] = (m^2 - u) / 2;
        ScalarProduct[p2, k1] = (m^2 - u) / 2;
```

Matrix element squared; term |Ms|^2

```
In[13]:= coeffSS = e^4 / (s - m^2) ^2 / (16 m^2);
         traceSS = Tr[
           GA[μ] . (GS[p1] + GS[k1] + m) . GA[ν] . (GS[p1] + m) .
            GA[ν] . (GS[p1] + GS[k1] + m) . GA[μ] . (GS[p2] + m)]
         ξSS = Expand[traceSS /. {t → 2 * m^2 - s - u}]
```

Out[14]= $4\left(-6\, m^4 + 10\, m^2\, s + 4\, m^2\, t + 6\, m^2\, u - 2\, s\, u\right)$

Out[15]= $8\, m^4 + 24\, m^2\, s + 8\, m^2\, u - 8\, s\, u$

```
In[16]:= (* compare ξSS to B in equation (14.112) *)
         B = 8 * (4 m^4 + (s - m^2) * (m^2 - u) + 2 m^2 * (s - m^2));
         Expand[B]
         (* Thus ξSS = B *)
```

Out[17]= $8\, m^4 + 24\, m^2\, s + 8\, m^2\, u - 8\, s\, u$

Matrix element squared; term |Mu|^2

```
In[18]:= coeffUU = e^4 / (u - m^2)^2 / (16 m^2);
         traceUU = Tr[
           GA[ν].(GS[p1] - GS[k2] + m).GA[μ].(GS[p1] + m).
            GA[μ].(GS[p1] - GS[k2] + m).GA[ν].(GS[p2] + m)]
         ξUU = Expand[traceUU /. {t → 2 * m^2 - s - u}]
         (* Note that ξUU is ξSS with the exchange s and u *)
```

Out[19]= $4\left(-6\, m^4 + 6\, m^2\, s + 4\, m^2\, t + 10\, m^2\, u - 2\, s\, u\right)$

Out[20]= $8\, m^4 + 8\, m^2\, s + 24\, m^2\, u - 8\, s\, u$

We also need the interference terms

$$M_1\, M_2{}^* + M_1{}^*\, M_2$$

Interference terms

$$M_s M_u^* = \frac{e^4}{(s-m^2)(u-m^2)}\, \bar{u}_2 \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu u_1\, \epsilon_{2\mu} \epsilon_{1\nu}\; \bar{u}_1 \gamma^\alpha (\not{p}_1 - \not{k}_2 + m) \gamma^\beta u_2\, \epsilon_{2\alpha} \epsilon_{1\beta}$$

$$\overline{M_s M_u^*} = \frac{e^4}{(s-m^2)(u-m^2)} \frac{1}{4(2m)^2}\, \mathrm{Tr}\, \gamma^\mu (\not{p}_1 + \not{k}_1 + m) \gamma^\nu (\not{p}_1 + m) \gamma_\mu (\not{p}_1 - \not{k}_2 + m) \gamma_\nu (\not{p}_2 + m)$$

Matrix element squared; the interference terms

```
In[21]:= coeffSU  = e^4 / (s - m^2) / (u - m^2) / (16 m^2);
         traceSU = Tr[
           GA[μ].(GS[p1] + GS[k1] + m).GA[ν].(GS[p1] + m).
            GA[μ].(GS[p1] - GS[k2] + m).GA[ν].(GS[p2] + m)]
         ξSU = Expand[traceSU /. {t → 2 * m^2 - s - u}]
         (* Note that ξSU is real *)
```

Out[22]= $4\left(12\, m^4 - 2\, m^2\, s - 8\, m^2\, t - 2\, m^2\, u + 2\, s\, t + 2\, t^2 + 2\, t\, u\right)$

Out[23]= $16\, m^4 + 8\, m^2\, s + 8\, m^2\, u$

Summing Up

```
In[24]:= MsqSS = coeffSS * ξSS;
         MsqUU = coeffUU * ξUU;
         MsqINT = 2 * coeffSU * ξSU;
         Msq = MsqSS + MsqUU + MsqINT;

In[28]:= (* Compare the Final Result to Equation (14.116) *)
         test = Msq /. {s → sM + m^2, u → uM + m^2};
         eq116 = (2 * e^4 / m^2) * ((m^2 / sM + m^2 / uM)^2 +
               (m^2 / sM + m^2 / uM) - 1 / 4 * (uM / sM + sM / uM));
         Simplify[test - eq116]
         (* Msq is same as equation (14.116) *)
```

Out[30]= 0

❖ The Klein-Nishina cross section

We'll calculate the cross section in the lab frame of reference, i.e., the rest frame of the initial electron.



$p_1^\mu = (m , 0, 0, 0)$; $k_1^\mu = (\omega , 0, 0, k)$

$k_2^\mu = (\omega_2 , k_2 \sin\theta , 0 , k_2 \cos\theta)$

$p_2^\mu = (E_2 , -k_2 \sin\theta , 0 , k - k_2 \cos\theta)$

$$k_1^\mu + p_1^\mu = k_2^\mu + p_2^\mu$$

Mandelstam variables in the lab frame

$$s = (k_1 + p_1)^2 = k_1^{\,2} + p_1^{\,2} + 2k_1.p_1$$
$$= m^2 + 2m\,\omega$$

$$u = (k_1 - p_2)^2 = (k_2 - p_1)^2$$
$$= m^2 - 2m\,\omega_2$$

$$t = (k_1 - k_2)^2 = k_1^{\,2} + k_2^{\,2} - 2\,k_1.k_2$$
$$= -\,2\,\omega\,\omega_2 + 2\,k\,k_2 \cos\theta$$
$$= -\,2\,\omega\,\omega_2\,(1 - \cos\theta)$$
$$= (p_2 - p_1)^2 = 2m^2 - 2mE_2$$
$$= 2m\,[\,m - (\omega + m - \omega_2)\,]$$
$$= 2m\,(-\,\omega + \omega_2)\,;$$

thus (*the **Compton effect***) :

$$1/\omega_2 \;=\; 1/\omega + (1/m)\,(1 - \cos\theta)$$

The cross section …

Recall the general formula for 2-2 scattering

$$d\sigma = \frac{(2m)^{n_f}}{4E_1E_2\, v_{rel}}(2\pi)^4\,\delta^4(P_f - P_i)\,\frac{d^3p_3}{(2\pi)^3\,2E_3}\,\frac{d^3p_4}{(2\pi)^3\,2E_4}\,|\mathcal{M}|^2$$

For Compton scattering, CoM frame,

$$d\sigma = \frac{(2m)^2}{4\omega m c}\,\frac{1}{(2\pi)^2}\int \delta(\omega_2 + E_2 - \omega - m)\,\frac{\overbrace{d^3k_2}^{k_2^2\,dk_2\,d\Omega_2}}{4\omega_2 E_2}\,|\mathcal{M}|^2$$

$$E_2 = \sqrt{m^2 + (-k_2\sin\theta)^2 + (k - k_2\cos\theta)^2}$$

$$= \sqrt{m^2 + k_2^2 + k^2 - 2k k_2\cos\theta}$$

$$\delta[f(x)] = \frac{\delta(x - x_0)}{|f'(x_0)|}$$

$$f(k_2) = k_2 + \sqrt{m^2 + k_2^2 + k^2 - 2k k_2\cos\theta}$$

$$f'(k_2) = 1 + \frac{1}{2}\,\frac{1}{\sqrt{\cdots}}\,(2k_2 - 2k\cos\theta)$$

$$= 1 + \frac{k_2 - k\cos\theta}{E_2} = \frac{1}{E_2}\{E_2 + k_2 - k\cos\theta\}$$

$$= \frac{1}{E_2}\left(m + k(1 - \cos\theta)\right)$$

$$\frac{d\sigma}{d\Omega_2} = \frac{(2m)^2}{4\omega m \underset{c=1}{\cancel{c}}}\,\frac{1}{(2\pi)^2}\,\frac{k_2^2}{4\omega_2 E_2}\,\frac{E_2}{m + \omega(1-\cos\theta)}\,|\mathcal{M}|^2$$

Compton effect : $\frac{m\omega}{\omega_2} = m + \omega(1 - \cos\theta)$

$$\frac{d\sigma}{d\Omega_2'} = \frac{1}{16\pi^2}\,\frac{\omega_2^2}{\omega^2}\,|\mathcal{M}|^2 \quad \text{where } |\mathcal{M}|^2 = \frac{e^4}{2m^2}\left(\frac{\omega_2}{\omega} + \frac{\omega}{\omega_2} - \sin^2\theta\right)$$

$$\frac{d\sigma}{d\Omega_2'} = \frac{\alpha^2}{2m^2}\left(\frac{\omega_2}{\omega}\right)^2\left(\frac{\omega_2}{\omega} + \frac{\omega}{\omega_2} - \sin^2\theta\right) \qquad (14.136)$$

Or, let Mathematica do the algebra

$s = m^2 + 2m\,\omega$
$u = m^2 - 2m\omega_2$
$1/\omega_2 = 1/\omega + (1/m)\,(1 - \cos\theta)$

Klein Nishina Cross Section; cross section in the lab frame

There is a misprint in Equation (14.135); Equation (14.136) is correct.

```
In[532]:= Clear[cos]
S = m^2 + 2 * m * ω1;
U = m^2 - 2 * m * ω2;
Msq = Expand[Msq /. {s → S, u → U}]
temp = Expand[Msq / (e^4 / (2 m^2))];
temp2 = temp - ω2 / ω1 - ω1 / ω2;
temp2 = Expand[temp2 /. {1 / ω2 → 1 / ω1 + 1 / m * (1 - cos)}];
temp2 = Expand[temp2 /. {(1 / ω2)^2 → (1 / ω1 + 1 / m * (1 - cos))^2}];
Asq = (e^4 / (2 m^2) * (temp2 + ω2 / ω1 + ω1 / ω2))
Δσ = 1 / (16 π^2) (ω2 / ω1)^2 * Asq /. {e → Sqrt[4 * π * α]}
```

Out[535]= $\frac{e^4\,\omega2}{2\,m^2\,\omega1} + \frac{e^4\,\omega1}{2\,m^2\,\omega2} + \frac{e^4}{m\,\omega1} - \frac{e^4}{m\,\omega2} + \frac{e^4}{2\,\omega1^2} - \frac{e^4}{\omega1\,\omega2} + \frac{e^4}{2\,\omega2^2}$

Out[540]= $\frac{e^4\left(\frac{\omega2}{\omega1} + \frac{\omega1}{\omega2} - 1 + \cos^2\right)}{2\,m^2}$

Out[541]= $\frac{\alpha^2\,\omega2^2\left(\frac{\omega2}{\omega1} + \frac{\omega1}{\omega2} - 1 + \cos^2\right)}{2\,m^2\,\omega1^2}$

The Klein - Nishina formula

$$\frac{d\sigma}{d\Omega} = \frac{\alpha^2}{2m^2}\,\frac{\omega_2{}^2}{\omega^2}\left[\frac{\omega_2}{\omega} + \frac{\omega}{\omega_2} - \sin^2\theta\right]$$

The nonrelativistic limit

- $\omega << m$; photon energy is small
- $\omega_2 \approx \omega$; no frequency shift
- $d\sigma/d\Omega \approx \alpha^2/(2m^2)\,(1 + \cos^2\theta)$;
- $\sigma_{total} \approx 8\pi\,\alpha^2/(3m^2)$

i.e., it agrees with the classical theory of light scattering (Thomson scattering)



Homework problems

due Friday April 7

14. Maiani and Benhar, problem 14.4.1.
15. Maiani and Benhar, problem 14.4.2.
17. Maiani and Benhar, problem 14.4.3.
17. Mandl and Shaw, problem 8.7.